

Funkcionalna verifikacija hardvera

Vežba 5

Uvod u UVM metodologiju razvoja verifikacionih okruženja

Sadržaj

1	Uvod	4
1.1	Istorijat	4
2	UVM verifikaciono okruženje	5
2.1	Tipična struktura testbenča	5
2.2	UVM klasna hijerarhija	6
2.3	UVM faze	7
2.4	UVM <i>factory</i>	10
2.4.1	Registracija	10
2.4.2	Podrazumevane vrednosti konstruktora	11
2.4.3	Kreiranje komponenti i objekata	11
2.5	UVM poruke	12
2.5.1	<i>Severity</i>	12
2.5.2	<i>Verbosity</i>	13
3	DUT	14
4	<i>UVM TEST</i>	17
5	<i>Virtualni interfejs</i>	19
6	<i>Konfiguraciona baza podataka (eng. Configuration Database)</i>	20
7	Simulacija pomoću Xcelium Simulatora	23
7.1	Kompilacija	23
7.2	UVM <i>command line processor</i>	23
8	Simulacija pomoću VIVADO alata	25
8.1	Kompilacija	25
8.2	UVM <i>command line processor</i>	27
9	Zadaci	28
10	Appendix	29

Ova i naredne vežbe su posvećene univerzalnoj verifikacionoj metodologiji. U ovoj vežbi je dat kratak pregled metodologije i objašnjeni osnovni koncepti. Data je funkcionalna specifikacija dizajna koji će se verifikovati i opis transakcija. Naredne vežbe su posvećene pojedinim komponentama, njihovoj ulozi i načinu korišćenja.

1 Uvod

UVM (engl. *Universal Verification Methodology*) je standardizovana metodologija za funkcionalnu verifikaciju sa pomoćnom bibliotekom u SystemVerilog jeziku. Kreirala ju je firma Accellera u saradnji sa mnogim EDA prodavcima i klijentima sa željom da se postigne lakša ponovna upotreba testbenčeva i jednostavno kreiranje univerzalnih, visoko-kvalitetnih VIP-a (engl. *Verification Intellectual Property*). UVM je baziran na OVM-u (engl. *Open Verification Methodology*) i eRM-u (engl. *e Resuse Methodology*).

Jedna od glavnih karakteristika ove metodologije je UVC (engl. *Universal Verification Component*), odnosno univerzalne verifikacione komponente koje imaju istu strukturu (sadrže monitore, drajvere, sekvencere, ...) što omogućava lako korišćenje bilo kako nezavisne komponente ili kao deo većeg sistema.

Objektno-orijentisani dizajn, kao glavna karakteristika SystemVerilog jezika, omogućava lako kreiranje verifikacionih komponenti, dok veliki broj predefinisanih funkcija i taskova znatno ubrzava proces kreiranja testova.

UVM obezbeđuje *framework* za verifikaciju zasnovanu na pokrivenosti (engl. *Coverage Driven Verification*, skraćeno CDV) koja ne zahteva kreiranje velikog broja testova, osigurava temeljnu verifikaciju na osnovu zadatih parametara i olakšava proces pronalaženja problema. Ovo se postiže korišćenjem *self-checking* testbenčeva, automatskog generisanja testova i korišćenjem podataka o pokrivenosti.

1.1 Istorijat


UVM 1.0 je nastao 2011 godine i, zbog velike podrške simulatora, ubrzo zamenio, do tada popularni, OVM. UVM 1.0 je preuzeo dosta iz OVM 2.1.1 verzije (u velikom delu koda samo zamenjujući prefikse "ovm_" sa "uvm_") i dodao par nadogradnji koje umnogome olakšavaju upotrebu i pružaju veću funkcionalnost. Među njima su *end-of-test* mehanizam za olakšavanje završavanja testova, *callback* mehanizam, lakša podešavanja izveštaja, ... UVM 1.0 je pratilo nekoliko manjih izdanja gde su ispravljene sitne greške i dodate nove funkcionalnosti uključujući registarski sloj, faze, sekvence, TLM interfejsa, ... Trenutna verzija UVM-a je UVM 1.2 koji je izašao 2014 godine. UVM biblioteka, kao i prethodne verzije se mogu preuzeti sa <http://www.accellera.org/downloads/standards/uvm>

2 UVM verifikaciono okruženje

U ovom poglavlju je dat kratak pregled klasičnog UVM testbenča. Ukratko je objašnjena struktura testbenča, navedene osnovne komponente, način kreiranja testova, ali i osnovni elementi UVM biblioteke. Dat je samo kratak pregled i uloga, a sve komponente i navedene UVM-ove mogućnosti će biti detaljno razrađene u narednim vežbama.

2.1 Tipična struktura testbenča

Svaki UVM testbenč se sastoji od određenog broja verifikacionih komponenti sa jasno definisanom ulogom, koje su povezane na odgovarajući način, prate metodologiju i omogućavaju laku ponovnu upotrebu. Na blok dijagramu (slika 1) je prikazana tipična struktura testbenča.




Slika 1: UVM testbenč

Prikazani delovi testbenča su:

- *data objekat* (paket, *sequence item*, transaction) - koristi se kao stimulus za dizajn koji se verifikuje. Mora sadržati polja koja su potrebna drajveru da bi uspešno inicirao transakciju, ali može sadržati i neka dodatna polja koja se koriste za kontrolu, kao i ograničenja pri randomizaciji
- sekvencer komponenta - je stimulus generator. U njemu se kontrolišu podaci koje će drajver slati DUT-u. Kontrola podataka se vrši u definisanim sekvencama koje određuju broj paketa, redosled, sadržaj i sl. Kreiranje sekvenci je glavni način generisanja stimulusa u UVM-u
- drajver komponenta - je aktivna komponenta koja emulira signale koji se šalju dizajnu. Na osnovu podataka od sekvencera i semplovanja signal na interfejsu, generiše signale na ulaze dizajna koji se verifikuje, prateći implementirani protokol
- monitor komponenta - je pasivna komponenta zadužena za nadgledanje signala, ali ne i njihovo generisanje. Može sakupljati podatke o pokrivenosti, ali i raditi proveru. Sakupljene informacije šalje ostalim komponentama na dalju obradu (npr. *scoreboard*-u na dalju kontrolu)

- agent komponenta - obuhvata drajver, sekvencer i monitor. Iako se ove podkomponente mogu koristiti nezavisno to iziskuje poznavanje imena, uloge i načina povezivanja. Agent obuhvata ove tri komponente, predstavlja apstraktniji pristup datom interfejsu i olakšava ponovno korišćenje ovih komponenti. Aktivan agent instancira sve tri komponente, dok pasivan instancira samo monitor (pasivan agent ne generiše signale, služi samo za nadgledanje; uglavnom je deo većeg okruženja, gde se generisanje podataka vrši na nekom višem nivou)
- *scoreboard* - na osnovu podataka dobijenih iz monitora vrši proveru
- konfiguracioni objekat - sadrži sve podatke potrebne za konfigurisanje testbenča npr. broj agenata, da li je potreban master ili *slave*, da li je agent aktivan ili pasivan i sl.
- okruženje - obuhvata sve prethodno navedene delove u jednu klasu radi jednostavnog korišćenja
- test

Hijerarhija testbenča je određena nizom *has-a* klasnim vezama tj. komponente na višem nivou hijerarhije sadrže komponente na nižem nivou. Npr. agent sadrži drajver, sekvencer i monitor. Ovo je prikazano na slici 2.




Slika 2: UVM hijerarhija

Sa ove slike se vidi i da je test na najvišem nivou hijerarhije. On je zadužen za instanciranje i konfiguraciju okruženja i startovanje sekvenci. Često postoji bazna test klasa u kojoj se vrše instanciranje okruženja, dok se u izvedenim testovima vrše specifične operacije za taj test npr. redosled sekvenci, randomizacija i sl.

2.2 UVM klasna hijerarhija

Na slici 3 je prikazana delimična UVM klasna hijerarhija. Nasleđivanje omogućava brz razvoj dobro strukturiranih verifikacionih komponenti i objekata. Bazine klase sadrže potrebna polja i metode za kontrolu, pa je, umesto pisanja svake komponente od nule, potrebno samo modifikovati određene metode za konkretne potrebe.

Kao što se vidi sa slike 3, svaki deo testbenča opisan u prethodnom poglavlju ima odgovarajuću UVM klasu i kako bi se implementirao potrebno je naslediti tu klasu. Npr. ukoliko želimo da implementiramo monitor, potrebno je napisati novu klasu koja nasleđuje *uvm_monitor*. Slično će drajver nasledivati *uvm_driver*, sekvencer *uvm_sequencer* i sl.



Slika 3: UVM klasna hijerarhija

Takođe se može uočiti da sve klase nasleđuju `uvm_object` klasu. Jedna od bitnijih podklasa je `uvm_component`. Klasa komponenta je izvedena iz klase objekat, međutim sadrži veoma bitne osobine zbog kojih se često posebno izdvaja. Prvenstveno sadrži UVM-ov mehanizam faza, mogućnost korišćenja konfiguracija i TLM interfejsa. Sve klase koje se koriste za kreiranje komponenti verifikacionog okruženja (`uvm_driver`, `uvm_monitor`, `uvm_env`, ...) nasleđuju `uvm_component` klasu. Obično se u literaturi kao dve glavne grupe navode objekti i komponente, pri čemu se pravi razlika između klasa nasleđenih iz `uvm_object` i klasa nasleđenih iz `uvm_component` (iako su i ove klase naseđene iz `uvm_object`, izdvajaju se zbog velikog broja posebnih osobina).

2.3 UVM faze

UVM-ov mehanizam faza je jedna od najbitnijih funkcionalnosti i novina u ovoj biblioteci. Kao što je već navedeno, jedna od osnovnih mogućnosti `uvm_component` klase su upravo faze. Pošto u verifikacionom okruženju postoji veliki broj komponenti koje rade u paraleli, potrebno je obezbediti da se određene funkcionalnosti izvrše u jasno definisanim trenucima. Faze predstavljaju sinhronizacioni mehanizam za okruženje odnosno sve komponente su uvek sinhronizovane u pogledu faza.

Postoje tri osnovne grupe UVM faza:


- *build* faze - za kreiranje i konfigurisanje okruženja
- *run-time* faze - gde zapravo teče simulaciono vreme
- *clean up* faze - sakupljanje i analiza rezultata

Ove faze su zapravo samo virtuelne metode u `uvm_component` klasi, čijim se modifikacijama omogućava jasno definisan *flow* u okruženju. Korišćenjem faza se povećava mogućnost ponovne upotrebe komponenti jer je uloga svake faze, i funkcionalnosti koje se u njoj izvršavaju, jasno definisana. Na slici 4 su prikazane sve UVM faze i grupisane u tri osnovne grupe.

U tabeli 1 je dat kratak opis svih faza u redosledu izvršavanja.

Run faza ima veliki broj pod-faza kojima se može vršiti detaljnija sinhronizacija. One su date u tabeli 2.

UVM pruža velike mogućnosti pri korišćenju ovih faza. Kao što se vidi, *run-time* faze obuhvataju veliki broj pod-faza čija je preporučena uloga i redosled jasno definisan. Radi lakšeg prelaska



Slika 4: UVM faze

Faza	Opis
<i>build</i>	instanciranje podkomponenti, podešavanje konfiguracija, ...
<i>connect</i>	međusobno povezivanje komponenti, podešavanje interfejsa, ...
<i>end_of_elaboration</i>	prikaz topologije, naknadne izmene (u ovoj fazi je okruženje potpuno kreirano i povezano)
<i>start_of_simulation</i>	prikaz topologije, priprema za početak simulacije, podešavanja <i>run-time</i> konfiguracija, simulatora, ...
<i>run</i>	sama simulacija (generisanje stimulusa, provera, ...)
<i>extract</i>	preuzimanje preostalih podataka iz komponenti (npr. informacije iz <i>scoreboard</i> -a ili monitora)
<i>check</i>	preostale provere
<i>report</i>	prikaz rezultata ili upis rezultata u fajlove
<i>final</i>	završetak svih preostalih radnji (npr. zatvaranje fajlova ili terminacija simulatora)

Tabela 1: UVM faze

Faza	Opis
<i>pre_reset</i>	sve aktivnosti koje treba obaviti pre reseta
<i>reset</i>	specifično ponašanje tokom reseta
<i>post_reset</i>	sve aktivnosti koje treba obaviti nakon reseta
<i>pre_configure</i>	priprema DUT-a za konfiguraciju posle reseta
<i>configure</i>	konfigurisanje DUT-a
<i>post_configure</i>	čekanje da DUT odreaguje na konfiguraciju
<i>pre_main</i>	provera da su sve komponente spremne za rad
<i>main</i>	glavna faza - generisanje stimulusa, sakupljanje podataka, čekiranje, ...
<i>post_main</i>	finalizacija aktivnosti iz <i>main</i> faze
<i>pre_shutdown</i>	pomoćna faza za aktivnosti pre <i>shutdown</i> faze
<i>shutdown</i>	provera da je generisan stimulus korektno propagiran, provera statusnih registara, ...
<i>post_shutdown</i>	finalizacija aktivnosti, poslednja aktivna faza

 Tabela 2: UVM *run* faze

sa OVM-a na UVM, omogućeno je korišćenje i samo *run* faze umesto svih pod-faza (pogledati sliku 4, *run* obuhvata *reset*, *main*, *post_main*, ..).

Dve najčešće korišćene faze su *build* i *run* (ili *main*), pa će kostur većine komponenti biti (primetiti da je *run/main* faza implementirana kao task zato što troši simulaciono vreme):

```
class example extends uvm_component;

    // polja i podkomponente
    // ...

    `uvm_component_utils(example) // factory registration

    function new(string name = "example", uvm_component parent);
        super.new(name, parent);
    endfunction : new

    function void build_phase(uvm_phase phase);
        super.build_phase(phase);
        // ...
    endfunction : build_phase
```

```

task main_phase(uvm_phase phase);
    super.main_phase(phase);
    // ...
endtask : main_phase

endclass : example

```

2.4 UVM *factory*

Factory metoda je jedna od klasičnih softverskih dizajn šablona (engl. *desing pattern*). Prilikom funkcionalne verifikacije često je potrebno uvesti promene u određenim klasama. Ovo se može desiti zbog uvođenja nove funkcionalnosti, specifičnih zahteva nekog testa i sl. Na primer, možda je potrebno dodati nova polja u transakciju ili modifikovati način generisanja signala ili neka ograničenja pri randomizaciji. UVM *factory* omogućava jednostavan način zamene postojeće klase izvedenom klasom bez menjanja ostatka okruženja. Primer: ukoliko u okruženju koristimo *old_transaction* klasu za slanje podataka, a želimo da koristimo *new_transaction* koja ima dodata određena ograničenja za randomizaciju potrebna za dati test. Ovo je moguće postići jednom linijom koda. Tada se sve instance *old_transaction* klase zamenjuju sa *new_transaction*, bez menjanja ostatka koda. Odnosno u celom okruženju se koristi *new_transaction* umesto *old_transaction*. Ovaj mehanizam se naziva *factory override* i biće obrađen u narednim vežbama.

Kako bi ovo bilo moguće, potrebno je pratiti određena pravila prilikom implementacije klasa, objašnjena u nastavku. Konvencija je uvek pisati kod na ovaj način radi lakše ponovne upotrebe.

2.4.1 Registracija

UVM objekti i komponente treba da sadrže kod za *factory* registraciju koji sadrži *wrapper* klasu oko date klase (mora biti *typedef*-ovana na *type_id*), statičnu funkciju da se dobije ovaj *type_id* i funkciju za dobijanje imena tipa. Međutim, pošto se pri svakoj registraciji moraju obaviti ovi koraci, koji prate dati šablon, razvijeni su određeni *factory registration* makroi. Konvencija je koristiti ove makroe umesto ručnog pisanja potrebnog koda kako bi se izbegle eventualne greške i povećala čitljivost koda. Postoje četiri makroa:

- ``uvm_component_utils(<name>)` - za komponente
- ``uvm_component_param_utils(<name>)` - za parametrizovane komponente
- ``uvm_object_utils(<name>)` - za objekte
- ``uvm_object_param_utils(<name>)` - za parametrizovane objekte

Ukoliko je potrebno i korišćenje *field* makroa, umesto ovih koriste se *begin..end* makroi koji kreiraju blok unutar kojeg se navode *field* makroi. Npr.

```

`uvm_object_utils_begin(<tip>)
  `uvm_field_* macro invocations here
`uvm_object_utils_end

```

Sami *field* makroi (makroi za polja u klasi) ne služe za registraciju već za implementaciju pomoćnih metoda (*copy*, *compare*, *print*, *pack*, ...). Njihova upotreba je opciona i biće demonstrirana u poglavlju o transakcijama.

Spisak svih *utility* makroa se može pronaći na https://verificationacademy.com/verification-methodology-reference/uvm/docs_1.2/html/files/macros/uvm_object_defines-svh.html

2.4.2 Podrazumevane vrednosti konstruktora

Pošto su konstruktori u *uvm_object* i *uvm_component* klasi zapravo virtuelne funkcije, potrebno je pratiti njihov prototip. Kako bi se omogućila ispravna konstrukcija, konstruktor treba da sadrži podrazumevane vrednosti za sve argumente.

Postoji razlika između komponenti i objekata:

```
// komponenta
function new(string name = "example ", uvm_component parent = null);
  super.new(name, parent);
endfunction

// objekat
function new(string name = "example");
  super.new(name);
endfunction
```

2.4.3 Kreiranje komponenti i objekata

Kako bi *factory* mogao da ispravno kreira objekte i komponente, potrebno je koristiti *create* metodu umesto direktnog pozivanja *new*. Npr.

```
class top_component extends uvm_component;

  sub_component comp; // some uvm_component
  sub_object obj; // some uvm_object

  `uvm_component_utils (top_component)

  function new(string name = "top_component", uvm_component parent = null);
    super.new(name, parent);
  endfunction : new

  function void build_phase(uvm_phase phase);
    super.build_phase();
    comp = sub_component::type_id::create("comp", this);
    obj = sub_object::type_id::create("obj");
    // umesto comp = new; obj = new;
  endfunction : build_phase

endclass : top_component
```

Prateći prethodne konvencije, skeleton svake objekat klase će biti nalik:

```
class example_class extends uvm_object;

  `uvm_object_utils (example_class)

  function new(string name = "example");
    super.new(name);
    // ...
  endfunction

  // ...

endclass : example
```

Slično i za komponente, koristeći odgovarajuće makroe i podrazumevane vrednosti konstruktora.

Sam mehanizam *override*-ovanja će biti obrađen u narednim vežbama.

2.5 UVM poruke

Za ispis poruka u SystemVerilog-u, koristili smo veliki broj ugrađenih funkcija: `$display`, `$write`, `$error`, `$warning`, ... Iako veoma korisne, UVM ipak pruža mnogo veće mogućnosti pri ispisu poruka uključujući kontrolu ispisa pojedinih poruka i kontrolu same simulacije. Zbog toga je preporuka uvek koristiti UVM-ov mehanizam poruka. Korišćenje ovog mehanizma je jako pojednostavljeno i potrebno je samo pozvati određene makroe, kojima prosleđujemo željene argumente.

2.5.1 Severity

U zavisnosti od značaja same poruke, postoji izbor od nekoliko makroa. Oni su:

- `'uvm_info(string <id>, string <mssg>, <verbosity>)`
- `'uvm_warning(string <id>, string <mssg>)`
- `'uvm_error(string <id>, string <mssg>)`
- `'uvm_fatal(string <id>, string <mssg>)`

Argumenti su: `<id>` za identifikaciju klase koja vrši ispis poruke; obično se navodi ime komponente, `<mssg>` je string koji sadrži samu poruku. `<verbosity>` služi za podešavanje nivoa i objašnjen je u narednom poglavlju. Razlika između ovih makroa je u akciji koja se vrši njihovim pozivanjem (tabela 3).

Značaj	Akcija	Opis
UVM_INFO	ispis poruke ukoliko je podešen odgovarajući nivo	koristi se za ispis raznih poruka; moguće podešavanje nivoa
UVM_WARNING	ispis poruke	koristi se za ispis upozorenja
UVM_ERROR	ispis poruke i završetak simulacije ukoliko se prevaziđe max broj dozvoljenih <i>error</i> -a	koristi se za ispis pronađenih grešaka u DUT-u
UVM_FATAL	ispis poruke i završetak simulacije	koristi se ukoliko postoji neka greška (bilo u okruženju, bilo u DUT-u) zbog koje nije moguće nastaviti simulaciju

Tabela 3: UVM *severity*

Par primera:

```
'uvm_fatal("NO_CFG", {"Config object must be set for: ", get_full_name(), ".cfg"})
'uvm_error(get_type_name(), $sformatf("Incorrect value observed: expected %0d, received %0d", expected,
received))
'uvm_warning(get_type_name(), "Inputs should be held low during reset")
```

`get_type_name()` funkcija vraća ime tipa koji je poziva i često se koristi kao prvi argument odnosno ID poruke. `$sformatf` je SystemVerilog-ova funkcija za manipulaciju stringova, nalik sličnim C-ovskim funkcijama, dok se u *fatal* primeru koristi konkatencija stringova.

Još jedna prednost korišćenja ovih makroa je ispis na kraju simulacije koji prikazuje broj prikazanih poruka, gde lako možemo uočiti da li su pronađene greške tokom simulacije:

```
# ** Report counts by severity
# UVM_INFO:      5
```

```
# UVM_WARNING:    0
# UVM_ERROR:      0
# UVM_FATAL:      0
```

2.5.2 Verbosity

Kao što je već napomenuto, UVM pruža mehanizam kontrole ispisa poruka. Ovaj mehanizam je zasnovan na prosleđivanju određene vrednosti svakoj poruci i zatim ispisu samo onih poruka sa odgovarajućim nivoima. Ovi predefinisani nivoi se mogu koristiti na dva načina - za pojedinačne poruke ili za celokupne komponente. Mi ćemo se zadržati na porukama.

Verbosity se podešava preko nekoliko predefinisanih *enum* vrednosti. One su prikazane u tabeli 4.

Nivo	Vrednost
UVM_NONE	0
UVM_LOW	100
UVM_MEDIUM	200
UVM_HIGH	300
UVM_FULL	400
UVM_DEBUG	500

Tabela 4: UVM *verbosity*

Kada setujemo nivo u testu, ispisivaće se sve poruke koje imaju taj ili niži nivo. Podrazumevani nivo je UVM_MEDIUM odnosno ispisivaće se poruke sa UVM_MEDIUM, UVM_LOW i UVM_NONE, odnosno filtriraće se sve poruke sa nivoom UVM_HIGH, UVM_FULL i UVM_DEBUG. Ako odaberemo UVM_FULL nivo, ispisivaće se sve poruke osim onih sa UVM_DEBUG nivoom.

Ispis UVM_NONE poruka se ne može isključiti.

Prilikom pisanja informacija treba koristiti ove nivoe na efikasan način, odnosno poruke koje služe samo za testiranje okruženja treba da imaju visoku vrednost, dok one koje će uvek biti od interesa treba da imaju nižu vrednost odnosno da se skoro uvek ispisuju.


Setovanje nivoa tokom simulacije je objašnjeno u poslednjem poglavlju, a u nastavku je dato par primera poruka sa različitim nivoima:

```
`uvm_info(get_type_name(), "Reset observed", UVM_MEDIUM)
`uvm_info(get_type_name(), "Configuration written", UVM_LOW)
`uvm_info(get_type_name(), "Collected transaction", UVM_HIGH)
```

3 DUT

U ovoj i narednim vežbama ćemo se baviti verifikacijom “Calc1” dizajna. Sve komponente će biti razvijane za ovaj primer. Ovaj dizajn se koristio i kao primer na predavanjima, dok ćemo na vežbama izvršiti kompletnu verifikaciju, prateći UVM metodologiju. U nastavku je data funkcionalna specifikacija DUT-a.

“Calc1” je kalkulator sa četiri porta koji podržava četiri operacije. Na svakom portu se može slati po jedna komanda koja može biti sabiranje, oduzimanje, šiftovanje u levo i desno. Pored komande, šalju se i prateći podaci. Pošto postoje četiri porta, mogu se postojati četiri aktivne komande u jednom trenutku. Komande i podaci se šalju u istom taktu, dok može proći nekoliko taktova pre dobijanja rezultata. Svaka komanda će dobiti odgovor i rezultat ukoliko nije došlo do greške. Do god se ne očita rezultat, nije dozvoljeno slanje nove komande. DUT sadrži dve aritmetičko-logičke jedinice, jedna procesira sve komande sabiranja, a druga je zadužena za komande šiftovanja. Prioritetnom logikom se šalju komande ka aritmetičko-logičkim jedinicama.



Slika 5: DUT

U narednoj tabeli je opisan interfejs našeg DUT-a.

Kao što se vidi u tabeli 5, svaki od četiri porta ima po dva ulazna i dva izlazna signala. *reqX_cmd_in* signal je četvorobitni ulazni signal za slanje komande. U tabeli 6 su prikazane moguće komande i njihova vrednost.

reqX_data_in je 32-bitni ulaz za podatke. Dva podatka koji predstavljaju operande se šalju na ovoj liniji, pri čemu se prvo šalje operand 1, a zatim operand 2 u sledećem taktu. Operand 1 se šalje u istom taktu kada i komanda. I odavde vidimo da je potrebno dva takta da bi se poslao zahtev DUT-u. U zavisnosti od komande, ova dva operanda se tumače na različite načine, koji su prikazani u tabeli 7.

Postoje i dve izlazne linije za svaki port. *out_respX* je dvobitni signal za odgovor koji daje informacije o uspešnosti operacije. Vrednosti su:

Rezultat operacije se nalazi na *out_dataX* liniji i validan je jedino u ciklusima u kojima je $out_respX = 2'b01$, odnosno kada je operacija uspešna.

Na slici 6 je prikazana uspešna operacija na portu 1. Komanda i prvi operand se šalju u jednom

Ime	Širina (br. bita)	Smer	Opis
c_clk	1	in	takt
reset	8	in	reset
reg1_cmd_in	4	in	komanda za port 1
reg1_data_in	32	in	podaci za port 1
reg2_cmd_in	4	in	komanda za port 2
reg2_data_in	32	in	podaci za port 2
reg3_cmd_in	4	in	komanda za port 3
reg3_data_in	32	in	podaci za port 3
reg4_cmd_in	4	in	komanda za port 4
reg4_data_in	32	in	podaci za port 4
out_resp1	2	out	odgovor za port 1
out_data1	32	out	rezultat za port 1
out_resp2	2	out	odgovor za port 2
out_data2	32	out	rezultat za port 2
out_resp3	2	out	odgovor za port 3
out_data3	32	out	rezultat za port 3
out_resp4	2	out	odgovor za port 4
out_data4	32	out	rezultat za port 4

Tabela 5: DUT interfejs

Komanda	Vrednost
<i>No operation</i>	4'b0000
<i>Add</i>	4'b0001
<i>Subtract</i>	4'b0010
<i>Shift left</i>	4'b0101
<i>Shift right</i>	4'b0110
<i>Invalid</i>	sve preostale vrednosti

Tabela 6: DUT komande


Komanda	Tumačenje operanada
<i>Add</i>	Rezultat je operand1 + operand2
<i>Subtract</i>	Rezultat je operand1 - operand2
<i>Shift left</i>	Rezultat je operand1 šiftovan u levo za operand2 mesta, pri čemu se koriste samo 5 bita drugog operanda. Uvek se ubacuju nule.
<i>Shift right</i>	Rezultat je operand1 šiftovan u desno za operand2 mesta, pri čemu se koriste samo 5 bita drugog operanda. Uvek se ubacuju nule.

Tabela 7: DUT operandi

Vrednost	Značenje
2'b00	Nema odgovora u ovom ciklusu.
2'b01	Uspešna operacija. Rezultujući podaci se nalaze na <i>out_dataX</i> liniji.
2'b10	<i>Overflow</i> , <i>underflow</i> ili pogrešna komanda. <i>Overflow/underflow</i> je moguć samo za operacije sabiranja i oduzimanja.
2'b11	Ne koristi se.

Tabela 8: DUT odgovor

ciklusu, a drugi operand se šalje u narednom ciklusu. Nakon nekoliko taktova dobija se odgovor i rezultat u istom ciklusu. Broj ciklusa koji je potreban da se dobije odgovor zavisi od aktivnosti na ostalim portovima (pošto postoje samo dve ALU), ali će uvek biti minimum 3 ciklusa.



Slika 6: Primer operacije na portu 1

Na svakom portu može biti aktivna samo jedna operacija, odnosno kada se pošalje jedna komanda, sledeća se ne sme slati do god ne primimo odgovor za prvu komandu. Takođe, slanje nove komande je opciono tj. port može biti neaktivan.

Četiri porta su nezavisna. Mogu se paralelno slati komande na svim ili samo nekim portovima, što znači da u bilo kom trenutku DUT može obavljati maksimum 4 operacije. Međutim, brzina odgovora će zavistiti od aktivnosti zbog ograničenih resursa u dizajnu. Kako što je već rečeno, DUT sadrži dve ALU, jednu za sabiranje i oduzimanje, a drugu za šiftovanje. Ukoliko se primi više komandi u paraleli za jednu ALU, DUT će ih serijalizovati i obrađivati jednu po jednu. Svaki port ima isti prioritet. DUT će obrađivati komande po principu *first-come, first-serve*, pri čemu za komande koje stignu u istom taktu nije definisan redosled.

Što se tiče reseta, aktiviranje se vrši postavljanjem reset linije na vrednost 8'b11111111. Ova vrednost se mora držati 7 sukcesivnih taktova kako bi se reset uspešno propagirao kroz dizajn. Tokom reseta, svi ulazi, osim takta, treba da budu nula.

U "Calc1" dizajnu se sve aritmetičke operacije vrše nad *unsigned* podacima. Prilikom sabiranja *overflow* se dešava ukoliko najviši bit ima *carry-out*. Slično *underflow* se pojavljuje prilikom oduzimanja ukoliko se oduzima veći broj od manjeg broja. U nastavku je dato par primera.

Komanda	Operand 1	Operand 2	Odgovor	Rezultat
<i>Add</i>	32'h80002345	32'h00010000	Uspešna operacija	32'h80012345
<i>Add</i>	32'hFFFFFFFF	32'h00000001	<i>Overflow</i>	Nema
<i>Subtract</i>	32'hFFFFFFFF	32'h11111111	Uspešna operacija	32'hEEEEEEEE
<i>Subtract</i>	32'h11111111	32'h20000000	<i>Underflow</i>	Nema

Tabela 9: DUT primeri operacija

4 UVM TEST

Kao što je već navedeno, na ovim vežbama ćemo započeti pravljenje verifikacionog okruženja za proveru "Calc1" dizajna. Prva komponenta verifikacionog okruženja zasnovanog na UVM metodologiji koju ćemo napraviti zove se *test*. Na slici 1 je prikazano da se ona u hijerarhiji nalazi direktno ispod *top* fajla. Sledeći kodni listing prikazuje kostur *test* komponente:

```

`ifndef TEST_SIMPLE_SV
`define TEST_SIMPLE_SV

class test_simple extends uvm_test;

    `uvm_component_utils(test_simple)

    function new(string name = "test_simple", uvm_component parent = null);
        super.new(name,parent);
    endfunction : new

    function void build_phase(uvm_phase phase);
        super.build_phase(phase);
        `uvm_info(get_type_name(), "Starting build phase...", UVM_LOW)
        // ...
    endfunction : build_phase

    task main_phase(uvm_phase phase);
        super.main_phase(phase);
        `uvm_info(get_type_name(), "Starting main phase...", UVM_LOW)
        // ...
    endtask : main_phase

endclass : test_simple

`endif

```

Kod 1: test_komponenta

Iz prethodnog se može videti da klasu *test* pored konstruktora čine još dve metode: *build_phase* i *main_phase*. Ovakav *Test* implementira podskup svih UVM faza, ostale se mogu ali i ne moraju implementirati.

Test komponenta se nalazi direktno ispod *Top-a*, stoga se tu i instancira. Kako bi se to omogućilo neophodno je uraditi sledeće u *Top* fajlu:

```

module calc_verif_top_run_test;

    //Ukljucivanje UVM biblioteka
    `include "uvm_macros.svh"
    import uvm_pkg::*;

    // Ukljucivanje test klase
    `include "v5_calc_test_simple.sv"
    // Pokretanje testa
    initial begin
        run_test("test_simple");
    end
endmodule : calc_verif_top_run_test
//Rezultat_izvršavanja
//
// UVM_INFO: Starting build phase...
// UVM_INFO: Starting main phase...

```

Kod 2: run_test

Funkcija `run_test` instancira objekat test klase i pokreće izvršavanje UVM faza. Kako bi `run_test` metoda znala koji test da pokrene neophodno joj je proslediti ime testa (kao parametar funkcije) ili podesiti simulator tako da sam zna koji test da pokrene(7). U prethodnom primeru metodi `run_test` je prosleđeno ime testa preko parametra. Rezultat izvršavanja prethodno navedenog koda jeste ispisivanje dve poruke iz *build* i *main* faza.

Prethodni primer je samo pokazao kako da se pokrene verifikaciono okruženje zasnovano na UVM metodologiji, ali da bi to okruženje radilo nešto smisleno neophodno je omogućiti mu pristup ulazima i izlazima komponente koja se verifikuje.

5 Virtualni interfejs

U vežbi 2 objašnjena je SystemVerilog konstrukcija koja se zove interfejs. Interfejsi služe za povezivanje dizajna i testbenča pri čemu je cilj je obuhvatiti svu komunikaciju na jednom mestu. Svi signali koji povezuju DUT i testbenč treba da se nalaze u interfejsima, čime se postiže jasna struktura koda. U primeru sa vežbe 2, gde se verifikovao jednostavni memorijski modul, svi signali povezani na ulaze i izlaze te memorije smešteni su u interfejs objekat. Ovim se postigla jasna struktura koda, ali takođe omogućena je još jedna stvar, lako prosleđivanje ovog interfejsa ostalim komponentama verifikacionog okruženja. U tom primeru, realizovana je drajver klasa kojoj se kao parametar konstruktora prosleđuje pokazivač na prethodno pomenuti interfejs:

```

`ifndef DRIVER_SV
`define DRIVER_SV

class driver;

    newTransaction tr;

    virtual memory_if mem_if;

    //Konstruktor
    function new(virtual memory_if mem_if);
        this.mem_if = mem_if;
    endfunction : new
    // Run metoda
    task run();
        tr = new();
        drive_transaction(tr);
        tr.addr = 2'hA;
        tr.en = 1'b1;
        drive_transaction(tr);
        tr.data_i = 8'hAA;
        tr.rw = 1'b1;
        drive_transaction(tr);
    endtask : run
    // Driver transaction metoda, postavlja vrednosti na ulaze memorije
    task drive_transaction(newTransaction tr);
        $display("Driving transaction:");
        tr.display_transaction();
        @(posedge mem_if.clk);
        mem_if.addr <= tr.addr;
        mem_if.data_i <= tr.data_i;
        mem_if.en <= tr.en;
        mem_if.rw <= tr.rw;
    endtask : drive_transaction

endclass : driver

`endif

```

Kod 3: v2_driver

Zbog ovoga, drajver klasa može da postavlja vrednosti na ulaze memorije koja se verifikuje, što se vidi u `drive_transaction` metodi. Da bi se naznačilo da je parametar pokazivač na pravi interfejs objekat iskorišćena je ključna reč *virtual*.

6 Konfiguraciona baza podataka (eng. Configuration Database)

U prethodnoj sekciji je objašnjen jedan način na koji mogu da se proslede objekti klasa, interfejsa, struktura drugim delovima(komponentama) verifikacionog okruženja. Kod verifikacionih okruženja zasnovanih na UVM metodologiji to se radi pomoću konfiguracionih baza podataka (eng. Configuration Database).

Uvm_config_db je UVM-ov konfiguracioni mehanizam baziran na tipovima koji nudi mogućnost hijerarhijskog specificiranja konfiguracionih parametara na željene vrednosti. uvm_config_db može podešavati skalarne objekte, pokazivače na klase, redove, nizove, čak i virtualne interfejsse. Metode za setovanje odnosno preuzimanje vrednosti su uvm_config_db::set i uvm_config_db::get. Njihovi prototipi su:

```
void uvm_config_db#(type T = int)::set(uvm_component cntxt, string inst_name, string field_name, T value);
bit uvm_config_db#(type T=int)::get(uvm_component cntxt, string inst_name, string field_name, ref T value);
```

gde je:

- T - tip podatka koji se prosleđuje (skalarni objekat, red, klasa, virtualni interfejs, ...)
- Cntxt - hijerarhijska putanja za koju se podešava / preuzima vrednost
- Inst_name - hijerarhijska putanja koja limitira pristup bazi
- Field_name - labela preko koje se traži unos
- Value - vrednost koja će se upisati u bazu ili pročitati iz baze

Get metoda vraća 1 ukoliko je vrednost uspešno preuzeta iz baze, a 0 u suprotnom.

Kod verifikacionih okruženja baziranih na UVM metodologiji virtualni interfejs se uvek prosleđuje koristeći ovaj mehanizam. U top modulu se instancira i poveže interfejs sa dizajnom koji se verifikuje, a zatim se prosleđi ostalim podkomponentama koristeći *uvm_config_db*. Primer ispod demonstrira setovanje instance *calc_vif* u bazi uz labelu *calc_if* za globalni opseg. Prikazan je jedino deo koda koji demonstrira *config_db*.

```
module calc_verif_top;

// ..
`include "uvm_macros.svh" // Include the UVM macros
import uvm_pkg::*; // import the UVM library
// Ukljucivanje fajla u kome je definisana calc_test_simple_2 klasa
`include "v5_calc_test_simple2.sv"
// interface
logic clk, rst;
calc_if calc_vif(clk, rst);

DUT
calc_top DUT(
    .c_clk      ( clk ),
    .reset      ( rst ),
    .out_data1  ( calc_vif.out_data1 ),
    .out_data2  ( calc_vif.out_data2 ),
    .out_data3  ( calc_vif.out_data3 ),
    .out_data4  ( calc_vif.out_data4 ),
    .out_resp1  ( calc_vif.out_resp1 ),
    .out_resp2  ( calc_vif.out_resp2 ),
    .out_resp3  ( calc_vif.out_resp3 ),
    .out_resp4  ( calc_vif.out_resp4 ),
    .req1_cmd_in ( calc_vif.req1_cmd_in ),
    .req1_data_in ( calc_vif.req1_data_in ),
    .req2_cmd_in ( calc_vif.req2_cmd_in ),
    .req2_data_in ( calc_vif.req2_data_in ),
```

```

        .req3_cmd_in ( calc_vif.req3_cmd_in ),
        .req3_data_in ( calc_vif.req3_data_in ),
        .req4_cmd_in ( calc_vif.req4_cmd_in ),
        .req4_data_in ( calc_vif.req4_data_in )
    );

    initial begin
        // Ubacivanje interfejsa u konfiguracionu bazu podataka
        uvm_config_db#(virtual calc_if)::set(null, "*", "calc_if", calc_vif);
        //pokretanje test_simple_2 testa
        run_test("test_simple_2");
    end

    // ...

endmodule : calc_verif_top

```

Kod 4: Primer setovanja virtualnog interfejsa

U primeru ispod prikazana ja UVM Test klasa koja pristupa konfiguracionoj bazi podataka, preuzima pokazivač na interfejs i pomoću njega pristupa ulazima i izlazima komponente koja se verifikuje:

```

`ifndef TEST_SIMPLE_SV
`define TEST_SIMPLE_SV

class test_simple_2 extends uvm_test;

    `uvm_component_utils(test_simple_2)

    virtual interface calc_if vif;

    function new(string name = "test_simple_2", uvm_component parent = null);
        super.new(name,parent);
    endfunction : new

    function void build_phase(uvm_phase phase);
        super.build_phase(phase);

        `uvm_info(get_type_name(), "Starting build phase...", UVM_LOW)

        // Preuzimanje virtuelnog interfejsa iz konfiguracione baze podataka.
        if (!uvm_config_db#(virtual calc_if)::get(null, "*", "calc_if", vif))
            `uvm_fatal("NOVIF",{ "virtual interface must be set:",get_full_name(),".vif"})
        // ...
    endfunction : build_phase

    task main_phase(uvm_phase phase);
        super.main_phase(phase);
        phase.raise_objection(this); //objasnjenje u materijalu za vezbu 6
        `uvm_info(get_type_name(), "Starting main phase...", UVM_LOW)
        //postavljanje komande
        vif.req1_cmd_in = 4'b0001;
        //postavljanje prvog podatka
        vif.req1_data_in = 32'h0001;
        //Cekanje 1 takt
        @(posedge vif.clk);
        //Uklanjanje komande
        vif.req1_cmd_in = 4'b0000;
        //Postavljanje drugog podatka
        vif.req1_data_in = 32'h0002;
        #1000ns;
        phase.raise_objection(this); //objasnjenje u materijalu za vezbu 6
        // ...
    endtask : main_phase

endclass : test_simple_2

```

```
`endif
```

Kod 5: Primer getovanja virtuelnog interfejsa

Bitno je napomenuti da se unutar test komponente (koja je implementirana prateći UVM metodologiju) ne treba postavljati vrednosti na ulaze digitalnog sistema. To je ovde urađeno radi lakšeg objašnjenja virtuelnog interfejsa i konfiguracione baze podataka.

Takođe, u prethodnom kodnom listingu upotrebljene su dve metode: `raise_objection` i `drop_objection`. One su detaljno objašnjene u vežbi 6, sekcija: Mehanizam završetka testa.

7 Simulacija pomoću Xcelium Simulatora

7.1 Kompilacija

Prilikom kompajliranja koda koji sadrži UVM biblioteku, potrebno je uključiti direktorijum koji sadrži UVM kod. U slučaju računara koji se koriste na laboratorijskim vežbama taj direktorijum se nalazi na sledećoj putanji:

```
/eda/cadence/2019-20/RHELx86/XCELIUM_19.03.013/tools/methodology/UVM/CDNS-1.2/
```

Tako da se prilikom kompajliranja fajlova i pokretanja simulacije pomoću `xrun` komande mora ukucati sledeće:

```
xrun -uvm +"/eda/cadence/2019-20/RHELx86/XCELIUM_19.03.013/tools/methodology/UVM/CDNS-1.2/"
-access +rwc calc_verif_top.sv
```

Kako kucanje ove putanje svaki put prilikom kompajliranja zna da postane naporno, može se napraviti sistemska varijabla koja će u sebi sadržati tu putanju. Ono što je neophodno uraditi jeste u `.bashrc` fajl, koji se nalazi u `home` direktorijumu, ukucati sledeće:

```
export UVM_HOME="/eda/cadence/2019-20/RHELx86/XCELIUM_19.03.013/tools/methodology/UVM/
CDNS-1.2/"
```

Sada se `xrun` komanda može pokrenuti na sledeći način:

```
xrun -uvm +$UVM_HOME -access +rwc calc_verif_top.sv
```

Napomena: ova podešavanja se odnose na alat koji se koristi na vežbama. Kod novijih verzija alata podešavanja mogu biti drugačija.

7.2 UVM *command line processor*

Command line processor klasa pruža mogućnosti preuzimanja informacija koje se daju preko komandne linije prilikom poziva simulacije. Mogućnosti ovog procesora su mnogobrojne i izlaze iz opsega ovog kursa. Mi ćemo se zadržati na podešavanjima samo nekoliko UVM varijabli sa komandne linije koje umnogome olakšavaju simulaciju - podešavanje opširnosti ispisa poruka (*verbosity*) i definisanje testa koji se pušta. Za detaljnije objašnjenje rada `uvm_cmdline_processor` klase pogledati "UVM Users Guide", poglavlje 6.6.

Test se pušta pozivom `run_test(<test_name>)` rutine u top modulu. Tada se kreira instanca datog testa i pokreće se simulacija. Sledeći primer pokreće test pod nazivom `test_example`:

```
module top;

import uvm_pkg::*;
`include "uvm_macros.svh"

// instantiate the DUT, connect,
initial begin
    run_test("test_example");
end
endmodule : top
```

Mane ovakvog pristupa su očigledne - svaki put kada želimo da pustimo drugi test, mora se menjati sors kod. Zbog toga se u praksi uvek koristi komandna linija za izbor testa. Kada se `run_test` pozove bez argumenta, proverava se `+UVM_TESTNAME=<test_name>` opcija. Sada se ne mora menjati sors kod, odnosno može se puštati više testova bez ponovnog kompajliranja čime se znatno štedi na vremenu. Primer prosleđivanja testa preko komandne linije:

```
xrun -uvm +$UVM_HOME +UVM_TESTNAME=test_simple -access +rwc calc_verif_top.sv
```

+UVM_VERBOSITY opcija pruža mogućnost promene verbosity u čitavom okruženju odnosno kontrolu ispisa poruka. Nivoi su opisani u drugom poglavlju ove vežbe (UVM_LOW, UVM_HIGH, ...). Primer upotrebe:

```
xrun -uvm +$UVM_HOME +UVM_TESTNAME=test_simple +UVM_VERBOSITY=UVM_HIGH -access +  
rwc calc_verif_top.sv
```


+UVM_MAX_QUIT_COUNT opcija pruža mogućnost završetka simulacije kada se dostigne dati broj grešaka u dizajnu. Ova opcija je posebno korisna tokom puštanja regresije. Podrazumevana vrednost je -1 odnosno simulacija se nikad ne završava na osnovu broja grešaka. Pod brojem grešaka se podrazumeva broj prijavljenih *uvm_error*-a.

8 Simulacija pomoću VIVADO alata


8.1 Kompilacija

Verzije Vivada od 2019.2 pa naviše omogućavaju kompajliranje UVM biblioteka, no kao i kod drugih alata neophodno je izvršiti određena podešavanja.


Nakon kreiranja projekta, prateći sledeće slike, podesiti kompajler tako da koristi UVM biblioteku prilikom kompajliranja:



Slika 7: Odabir *settings* opcije



Slika 8: Podešavanje *xvlog compile* opcije



Slika 9: Podešavanje *xelab elaboration* opcije


Naravno ova podešavanja mogu da se izvrše i kucanjem *.tcl* komandi:

```
set_property -name {xsim.elaborate.xelab.more_options} -value {-L uvm} -objects [get_filesets sim_1]
set_property -name {xsim.compile.xvlog.more_options} -value {-L uvm} -objects [get_filesets sim_1]
```

Nakon što su ova podešavanja izvršena moguće je kompajlirati fajlove koji sadrže UVM konstrukcije u sebi.

8.2 UVM *command line processor*

Dodatno, kao i kod Xcelium Simulatora, moguće je podešavati određene UVM varijable (UVM_TEST, UVM_Verbosity, itd) i isto kao malopre, to može da se uradi iz komandne linije ili u GUI *modu*. Ukoliko se koristi GUI neophodno je uraditi sledeće:



Slika 10: Podešavanje UVM varijabli

Sa slike 10 može se videti da u polje `xsim.simulation.xsim.more_options*` treba da se upišu podešavanja varijabli i to na sledeći način:

```
-testplusarg UVM_TESTNAME=test_simple -testplusarg UVM_VERBOSITY=UVM_LOW
```

Takođe prethodno podešavanje je moguće izvršiti iz komandne linije:

```
set_property -name {xsim.simulate.xsim.more_options} -value {-testplusarg UVM_TESTNAME=test_simple  
-testplusarg UVM_VERBOSITY=UVM_LOW} -objects [get_filesets sim_1]
```

Napomena: Nemojte kopirati prethodne instrukcije, već ih prekucajte. Neki karakteri mogu biti pogrešno kopirani.

9 Zadaci

Zadatak1 U pratećim fajlovima za ovu vežbu je povezan naš DUT sa interfejsom i kreiran kostur za UVM test klasu. Proučiti date fajlove i način puštanja testova.

Zadatak2 Proširiti Test klasu prikazanu u kodnom listingu 9 tako da se ispituju sledeće funkcionalnosti calc modula:

- Osnovi protokol (komanda / odgovor) na svakom portu.
- Osnovne operacije svake komande na svakom portu.
- Overflow i underflow za operacije sabiranja i oduzimanja.
- Oduzimanje dva jednaka broja.
- Istovremeno izvršavanje operacija na svim portovima.

U dodatnom materijalu su priloženi kodovi za kalkulator kao i skelet verifikacionog okruženja.

10 Appendix

```

`ifndef CALC_IF_SV
`define CALC_IF_SV

interface calc_if (input clk, logic [6 : 0] rst);

    parameter DATA_WIDTH = 32;
    parameter RESP_WIDTH = 2;
    parameter CMD_WIDTH = 4;

    logic [DATA_WIDTH - 1 : 0] out_data1;
    logic [DATA_WIDTH - 1 : 0] out_data2;
    logic [DATA_WIDTH - 1 : 0] out_data3;
    logic [DATA_WIDTH - 1 : 0] out_data4;
    logic [RESP_WIDTH - 1 : 0] out_resp1;
    logic [RESP_WIDTH - 1 : 0] out_resp2;
    logic [RESP_WIDTH - 1 : 0] out_resp3;
    logic [RESP_WIDTH - 1 : 0] out_resp4;
    logic [CMD_WIDTH - 1 : 0] req1_cmd_in;
    logic [DATA_WIDTH - 1 : 0] req1_data_in;
    logic [CMD_WIDTH - 1 : 0] req2_cmd_in;
    logic [DATA_WIDTH - 1 : 0] req2_data_in;
    logic [CMD_WIDTH - 1 : 0] req3_cmd_in;
    logic [DATA_WIDTH - 1 : 0] req3_data_in;
    logic [CMD_WIDTH - 1 : 0] req4_cmd_in;
    logic [DATA_WIDTH - 1 : 0] req4_data_in;

endinterface : calc_if

`endif

```

Kod 6: calc_if

```

`ifndef V5_CALC_TEST_PKG_SV
`define V5_CALC_TEST_PKG_SV
package calc_test_pkg;
`include "uvm_macros.svh" // Include the UVM macros
import uvm_pkg::*; // import the UVM library
`include "calc_test_simple.sv"
endpackage
`endif;

```

Kod 7: calc_test_pkg

```

module calc_verif_top;

    import uvm_pkg::*; // import the UVM library
    `include "uvm_macros.svh" // Include the UVM macros

    import calc_verif_pkg::*;

    logic clk;
    logic [6 : 0] rst;

    // interface
    calc_if calc_vif(clk, rst);

    // DUT
    calc_top DUT(
        .c_clk      ( clk ),
        .reset      ( rst ),
        .out_data1  ( calc_vif.out_data1 ),
        .out_data2  ( calc_vif.out_data2 ),
        .out_data3  ( calc_vif.out_data3 ),
        .out_data4  ( calc_vif.out_data4 ),
    );
endmodule

```

```

        .out_resp1 ( calc_vif.out_resp1 ),
        .out_resp2 ( calc_vif.out_resp2 ),
        .out_resp3 ( calc_vif.out_resp3 ),
        .out_resp4 ( calc_vif.out_resp4 ),
        .req1_cmd_in ( calc_vif.req1_cmd_in ),
        .req1_data_in ( calc_vif.req1_data_in ),
        .req2_cmd_in ( calc_vif.req2_cmd_in ),
        .req2_data_in ( calc_vif.req2_data_in ),
        .req3_cmd_in ( calc_vif.req3_cmd_in ),
        .req3_data_in ( calc_vif.req3_data_in ),
        .req4_cmd_in ( calc_vif.req4_cmd_in ),
        .req4_data_in ( calc_vif.req4_data_in )
    );

// run test
initial begin
    uvm_config_db#(virtual calc_if)::set(null, "*", "calc_if", calc_vif);
    run_test("test_simple");
end

// clock and reset init .
initial begin
    clk <= 0;
    rst <= 1;
    #50 rst <= 0;
end

// clock generation
always #50 clk = ~clk;
endmodule : calc_verif_top

```

Kod 8: calc_verif_top

```

`ifndef TEST_SIMPLE_SV
`define TEST_SIMPLE_SV

class test_simple_2 extends uvm_test;

    `uvm_component_utils(test_simple_2)

    virtual interface calc_if vif;

    function new(string name = "test_simple_2", uvm_component parent = null);
        super.new(name,parent);
    endfunction : new

    function void build_phase(uvm_phase phase);
        super.build_phase(phase);

        `uvm_info(get_type_name(), "Starting build phase...", UVM_LOW)

        // Preuzimanje virtuelnog interfejsa iz konfiguracione baze podataka.
        if (!uvm_config_db#(virtual calc_if)::get(null, "*", "calc_if", vif))
            `uvm_fatal("NOVIF",{ "virtual interface must be set:",get_full_name(),".vif"})
        // ...
    endfunction : build_phase

    task main_phase(uvm_phase phase);
        super.main_phase(phase);
        phase.raise_objection(this); //objasnjenje u materijalu za vezbu 6
        `uvm_info(get_type_name(), "Starting main phase...", UVM_LOW)
        //postavljanje komande
        vif.req1_cmd_in = 4'b0001;
        //postavljanje prvog podatka
        vif.req1_data_in = 32'h0001;
    endtask

```

```
//Cekanje 1 takt
@(posedge vif.clk);
//Uklanjanje komande
vif.req1_cmd_in = 4'b0000;
//Postavljanje drugog podatka
vif.req1_data_in = 32'h0002;
#1000ns;
phase.raise_objection(this); //objasnjenje u materijalu za vezbu 6
// ...
endtask : main_phase
endclass : test_simple_2
`endif
```

Kod 9: test_simple